GLMM Short Course **Spring 2008**

LAB 2: Linear Mixed Models

The dataset for this lab reports weights of genetically engineered mice that are susceptible to tumor growth. Each mouse is weighed weekly from day 8 to 64. Group 1 is a control group, but the other four are treated with anti-tumor agents and differences in weights are interpreted mainly as differences in tumor weights. The first figure below shows the correlations between the weekly measurements. The second shows the log weights over time for the first two groups.

1) Fit a model with mouse as a random effect. What fixed effects do you need in the model? Is there an advantage to "centering" the day effect by using day-36 instead of day by itself?
2) Fit a model with a random intercept using the subject=mouse option.
3) How do the models in 1) and 2) compare?
4) Fit a model with random intercepts and random slopes for mice. How does this compare to the model in 2)? Which fits better?
5) Fit a model with random intercepts and an AR(1) error structure to the residuals. How does this compare to the models in 2) and 4)? Can you simplify the model?
6) Fit a model that allows separate AR(1) and residual error terms for each of the groups. Does this model fit better?
7) Finally – what model would you select and what group effects are there?

```
data one;
set work.tumorwght;
/*Create duplicate day so day can be categorical and continuous */
catday=day;
cday=day-36;
day2=day*day;
cday2=cday*cday;
run;
proc mixed covtest;
class group mouse;
model lnweight=day group group*day day2 group*day2/ddfm=kenrog;
random mouse;
run;
proc mixed covtest;
class group mouse;
model lnweight=day group group*day day2/ddfm=kenrog;
random mouse;
run;
proc mixed covtest;
class group mouse;
model lnweight=cday group group*cday cday2 group*cday2/ddfm=kenrog;
random mouse;
run;
proc mixed covtest;
class group mouse;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
random mouse;
run;
proc mixed covtest;
class group mouse;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
random intercept/subject=mouse;
run;
proc mixed covtest;
class group mouse;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
random intercept cday/subject=mouse type=un g;
run;
proc mixed covtest;
class group mouse catday;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
random intercept/subject=mouse g;
repeated catday/subject=mouse type=ar(1) r;
run;
proc mixed covtest;
class group mouse catday;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
repeated catday/subject=mouse type=ar(1) r;
run;
proc mixed covtest;
class group mouse catday;
model lnweight=cday group group*cday cday2/ddfm=kenrog;
repeated catday/subject=mouse type=ar(1) r group=group;
run;
```

```
library(nlme)
tumorw<-read.table("tumorwght.txt", header=TRUE)
tumorw<-fecfat[order(tumor[,"mouse"]),]
tumorw$groupf<-factor(tumorw$group)
#random intercepts
lmefit<-lme(lnweight ~ day + groupf, random=~1 | mouse, data=tumorw)
summary(lmefit)
#random intercepts and slopes
lmefitsl<-lme(lnweight ~ day + groupf, random=~ 1 + day | mouse, data=tumorw)
summary(lmefitsl)
#random intercepts and slopes, AR(1) errors
lmefitslAR<-lme(lnweight ~ day + groupf, random=~ 1 + day | mouse,
cor=corAR1(form=~day), data=tumorw)
summary(lmefitslAR)
```